
KAJ JE REKURZIJA?

   Rekurzija je lastnost posameznih objektov, da se definirajo s pomočjo samega sebe, oziroma, da sodelujejo v definiciji samega sebe. V programskem jeziku Pascal pa rekurzija omogoča, da funkcije in podprogrami kličejo samega sebe. Pri tem moramo paziti, da algoritem izpolnjuje naslednja pogoja:

1. opisan mora biti v končnem številu korakov,

2. vsebovati mora nek pogoj, kdaj naj se rekurzija konča, saj bi se sicer lahko klicala v neskončnost.

   Pogoj končanja izvajanja funkcije ali podprograma ponavadi dosežemo z uporabo IF  stavka. Rekurzijo bomo podrobneje spoznali na naslednjih primerih:

1. primer

Na primeru podprograma Levo in Desno razloži delovanje rekurzivnega klica v programskem jeziku Pascal.
Program Osnove;

var

n : integer;

Procedure Desno(n : integer);

{Ukaze podprograma (write) izvršimo pred rekurzivnim klicem.}

begin

   write(n:3);  

   if n > 0 then     {zaustavitveni pogoj}

   desno(n-1)        {rekurzivni klic podprograma z novimi parametri}

end;

Procedure Levo(n : integer);

{Ukaze podprograma (write) izvršimo šele, ko se končajo vsi rekurzivni klici.}

begin

   if n > 0 then      {zaustavitveni pogoj}

   levo(n-1);         {rekurzivni klic podprograma z novimi parametri}

   write(n:3);

end;

begin                                 {glavni program}

   write('Vnesite vrednost n ');

   readln(n);

   desno(n);

   writeln;

   levo(n);

end.

Podprogram Desno najprej izvrši ukaze (write), šele nato, če zaustavitveni pogoj dopušča,

se izvede rekurzivni klic. Z rekurzivnim klicem se gre ponovno na začetek podprograma,

vendar sedaj za neke nove vrednosti parametrov. Dobljeni izpis bo imel obliko n, n-1,

n-2,..1, 0.

V podprogramu Levo pa, če zaustavitveni pogoj dopušča,se najprej podatki zapišejo v sklad nato, šele nato pa se izvede rekurzivni klic. Z rekurzivnim klicem se gre ponovno na začetek podprograma, vendar sedaj za neke nove vrednosti parametrov. To se ponavlja, dokler ni izpolnjen zaustavitveni pogoj. Ko pogoj n > 0 ni izpolnjen, se izpiše vrednost za n = 0. Nato pa se izvedejo ostali ukazi, vendar za vrednosti, ki so zapisane na vrhu sklada. Dobljeni izpis bo imel obliko 0 1 2 3 ...n.

Za vajo reši naslednja primera:

1. Izračunaj faktorielo poljubnega naravnega števila n. Vemo, da faktoriela števila pomeni produkt 

(n * (n - 1) * (n - 2) *…1. 

Pomoč: Nalogo rešiš podobno, kot je zapisan podprogram Levo.

2. Izračunaj vsoto števil od 1 do n. Vsota je 1 + 2 + 3 + …+ (n -1)  + n. 

Pomoč: Pazi pri zaustavitvenem pogoju, da greš od 1 in ne od 0.

2. primer

Vremenoslovci dnevno merijo temperaturo ozračja. Vrednosti si zapisujejo v tabelo. Izdelaj program, ki bo s pomočjo rekurzivnega klica določil najvišjo in najnižjo temperaturo v določenem časovnem intervalu.

program max_min;

type podatki = array[1..100] of integer;

var

tab: podatki;

max,min,od,dok: integer;

procedure napolni_tabelo(var tab : podatki);

var

i, stevilo : integer;

begin

  write('Vnesite stevilo meritev (maksimalno 100) ');

  readln(stevilo);

  for i:=1 to stevilo do begin

    write(i:0,' - ta meritev je ');

    readln(tab[i]);

  end;

end;

procedure maxmin(t:podatki; od,dok:integer; var max,min : integer);

var

s : integer;

l_min, l_max, d_min, d_max : integer;

begin

  if od = dok then begin max := t[od]; min := max end

  else if od = dok - 1 then  {dva elementa}

    if  t[od] < t[dok] then begin max := t[dok]; min := t[od] end

    else begin max := t[od]; min:= t[dok] end

  else begin {več elementov}

    s := (od + dok) div 2;  {sredina}

    maxmin(t, od, s, l_max, l_min);

    maxmin(t, s + 1, dok, d_max, d_min);

    if l_max > d_max then max := l_max else max := d_max;

    if l_min < d_min then min := l_min else min := d_min

  end;

end;

begin {glavni program}

  napolni_tabelo(tab);

  write('Od katere meritve dalje iscem maksimum in minimum. ');

  readln(od);

  write('Do katere meritve iscem maksimum in minimum ');

  readln(dok);

  maxmin(tab,od,dok,max,min);

  writeln('Najvišja temperatura je ',max:0,'  najnižja pa ',min:0);

end.

Kako poteka izvajanje programa?

 Vnesene vrednosti naj bodo: 5, 3, 7, 1, 6, 4, 2. 

Spremenljivka od dobi vrednost 1, spremenljivka dok pa vrednost 7. V spremenljivki t so zapisane vrednosti: t[1] = 5, t[2] = 3, t[3] = 7, t[4] = 1, t[5] = 6, t[6] = 4 in t[7] = 2. 

ŠT.
OPIS


1. NIVO
2. NIVO
3. NIVO

1. 
Začne se izvajati podprogram maxmin. od = 1, dok = 7.



2. 
Pogojni stavek (if od = dok) ima nepravilno vrednost, zato se ne izvedejo stavki begin max := t[od];  min := max end; , temveč program nadaljuje s testiranjem naslednjega pogojnega stavka.



3. 
Tudi pogojni stavek (if od  = do - 1 …)  ima nepravilno vrednost. Program se nadaljuje na sestavljenem stavku begin s:= (od + dok) div 2.



4. 
 Spremenljivka  s = (1 + 7) div 2 = 4.



5. 
Na sklad se zapišejo vsi podatki polja t , 1, 7. 



6. 

Z rekurzivnim klicom maxmin(t, od, s, l_max, l_min)  gre na začetek podprograma, vendar je sedaj vrednost spremenljivke  od = 1, dok = s = 4


7. 

Pogojni stavek (if od = dok) ima nepravilno vrednost, zato se ne izvedejo stavki begin max := t[od];  min := max end; , temveč program nadaljuje s testiranjem naslednjega pogojnega stavka.


8. 

Tudi pogojni stavek (if od  = do - 1 …)  ima nepravilno vrednost. Program se nadaljuje na sestavljenem stavku begin s:= (od + dok) div 2.


9. 

Spremenljivka s = (1 + 4) div 2  = 2.


10. 


Na sklad se zapišejo vsi podatki polja t , 1, 4.

11. 


Z rekurzivnim klicom maxmin(t, od, s, l_max, l_min)    gre na začetek podprograma, vendar je sedaj vrednost spremenljivke  od = 1, dok = s = 2.

12. 


Pogojni stavek (if od = dok) ima nepravilno vrednost, zato se ne izvedejo stavki begin max := t[od];  min := max end; , temveč program nadaljuje s testiranjem naslednjega pogojnega stavka.

13. 


Pogojni stavek (if od  = do - 1 …)  ima pravilno logično vrednost, saj je 1 = 2 – 1. Spremenljivkama l_max in l_min procedura priredi vrednosti  l_max = t[od] = t[1] = 5 in l_min = t[dok] = t[2] = 3.

14. 


Sedaj se program izvede naprej in pridemo do drugega rekurzivnega klica maxmin(t, s + 1, dok, d_max, d_min), pri čemer so vrednosti za s = (1 + 3) div 2 = 2 in 

dok = 4.

15. 


Pogojni stavek (if od = dok) ima nepravilno vrednost, zato se ne izvedejo stavki begin max := t[od];  min := max end; , temveč program nadaljuje s testiranjem naslednjega pogojnega stavka.

16. 


Pogojni stavek (if od  = do - 1 …)  ima pravilno logično vrednost, saj je 3 = 4 – 1. Spremenljivkama d_max in d_min procedura priredi vrednosti  d_max = t[od] = t[3] = 7 in d_min = t[dok] = t[4] = 1.

17. 


Sedaj se lahko izvede primerjanje (if l_max > d_max then max:=…). S primerjanjem l-max in d_max ter l_min in d_min dobimo vrednosti za max = 7 in min = 1.

18. 

Vrnemo se na  mesto za klicem v levo, en nivo višje. Sedaj se izvede klic podprograma maxmin (t, s + 1, dok, d_max, d_min), pri čemer so vrednosti za s=4, dok =7.


19. 

Pogojni stavek (if od = dok) ima nepravilno vrednost, zato se ne izvedejo stavki begin max := t[od];  min := max end; , temveč program nadaljuje s testiranjem naslednjega pogojnega stavka.


20. 

Tudi pogojni stavek (if od  = do - 1 …)  ima nepravilno vrednost. Program se nadaljuje na sestavljenem stavku begin s:= (od + dok) div 2.




21. 

s= (5 + 7) div 2 =6


22. 


Rekurzivni klic maxmin(t, od, s, l-max, l_min), pri čemer je od = 5, s = 6. 

23. 


Pogojni stavek (if od = dok) ima nepravilno vrednost, zato se ne izvedejo stavki begin max := t[od];  min := max end; , temveč program nadaljuje s testiranjem naslednjega pogojnega stavka.

24. 


Pogojni stavek (if od  = do - 1 …)  ima pravilno logično vrednost, saj je 5 = 6 – 1. Spremenljivkama l_max in l_min procedura priredi vrednosti  l_max = t[od] = t[5] = 6 in l_min = t[dok] = t[6] = 4.

25. 


Izvede se klic maxmin(t, s+1, dok, d_max, d_min), za s = 6, dok = 7.

26. 


Pogojni stavek (if od = dok) ima pravilno vrednost, priredijo se vrednosti spremenljivkama d_max in d_min: 

d_max = 2, d_min = 2.

27. 

Izvede se primerjanje (if l_max > d_max then max:=…). S primerjanjem l-max in d_max ter l_min in d_min dobimo vrednosti za max = 6 in min = 2.


28. 
Sedaj se izvede primerjanje (if l_max > d_max then max:=…). S primerjanjem l-max in d_max ter l_min in d_min dobimo vrednosti za max = 7 in min = 1.



29. 
Konec podprograma.





                                         







KONEC                KONEC             KONEC                      KONEC                             

Nalogo lahko prikažemo tudi v obliki drevesa. V vozliščih imamo vrednosti spremenljivke od in dok . Modre povezave predstavljajo vstop v rekurzijo, zelene pa vračanje iz rekurzije. 



Za vajo reši primera:

1. V tabeli imamo n naključnih števil. Poišči vrednost najmanjšega in največjega števila.

2. V veleblagovnici imajo dnevno n nakupov. Poišči znesek najmanjšega in največjega nakupa. Tabelo vrednosti nakupov generiraj naključno. 

3.primer
Na palici je n diskov, razvrščenih od najmanjšega do največjega. Preložiti jih je treba na drugo palico tako, da nikoli ne damo večji disk na manjšega. Pri tem si lahko pomagamo še z eno pomožno palico, hkrati pa lahko premaknemo le en disk. 

                          A                       B                                    C


                                     Začetni položaj

                         A                       B                                   C


                                    Končni položaj




                   Dovoljen premik                            Prepovedan premik

Problem rešimo tako, da damo na palico C najprej zadnji, največji obroč. To pa lahko storimo šele tedaj, ko preložimo n – 1 obročev s palice A na palico B. In kako smo do tega položaja prišli' Tako, da smo rešili enak problem, kot je prvotni, le da za n – 1 obročev, ki jih prelagamo iz A na B s pomočjo C. Sedaj preložimo n – ti obroč z A na C. Sedaj pa moramo preložiti še  n – 1 obročev z B na C. To pa je problem enake vrste kot prej. Problem Preloži n obročev z A na C s pomočjo B rešimo v treh korakih:

1. Preloži n – 1 obročev z A na B s pomočjo C

2. Preloži obroč z A na C

3. Preloži n – 1 obročev z B na C s pomočjo A

Pri prelaganju se ustavimo, ko je število obročev enako 1 (n = 1) in moramo preložiti še zadnji obroč iz A na C.

program prelozi;

{Program preloži obroče iz prvega stebra preko vmesnega stebra na

tretji steber}

var

n : integer;

st1, st2, st3 : char;

procedure hanoi(n: integer; st1, st2, st3: char);

begin

  if (n = 1) then 

  writeln( 'Prelozi z ', st1,' na ', st3) {zaustavitveni pogoj}

  else

  begin

    hanoi(n - 1, st1, st3, st2); {rekurzivni klic}

    writeln('Prelozi z ', st1, ' na ', st3);

    hanoi(n - 1, st2, st1, st3)  {rekurzivni klic}

  end

end;

begin {glavni program}

  write(' Stevilo obrocev je ');

  readln(n); 

  st1 := 'A';

  st2 := 'B';

  st3 := 'C';

  hanoi(n, st1, st2, st3);

end.

Za vajo reši naslednja primera:

1. S pomočjo rekurzije poišči največji skupni  večkratnik števil a, b in c. A, b,c so naravna števila.. 

2. S pomočjo rekurzije ugotovi ali je n praštevilo.

4. primer

V razredu je n učencev. Na testu iz matematike so dobili ocene od 1 do 100 točk. Izdelajte program, ki bo razvrstil učence glede na doseženo število točk. Pomagamo si z algoritmom hitrega urejanja.  

Najprej bomo izbrali delilni element in tabelo razdelili na dva dela. V prvem delu bodo vsi elementi, ki so manjši ali enaki delilnemu elementu, v drugem pa vsi elementi, ki so večji ali enaki delilnemu elementu. Nato bomo s pomočjo rekurzivnih klicev uredili oba dela. 

Pri vsakem rekurzivnem klicu bodo v  prvem delu tabele  elementi, ki so manjši ali enaki trenutnemu delilnemu elementu, v drugem delu pa elementi, ki so večji ali enaki trenutnemu delilnemu elementu. S tem bomo dosegli, da bosta oba dela tvorila urejeno zaporedje kot celoto, saj bodo elementi prvega dela vsi manjši ali kvečjemu enaki kot elementi drugega dela.  Klici v globino se bodo izvajali tako dolgo, dokler ne bo v delu tabele, ki ga urejamo, ostal samo še en element. Če pa je v delu tabele samo  en element, je ta del tabele zagotovo urejen. Ker pa so urejeni tudi vsi vmesni deli (posamezna zaporedja na različnih nivojih), pomeni, da je celotna tabela tudi urejena. 

   Tabelo lahko razdelimo na različne načine, vendar je od načina delitve tudi odvisna hitrost delovanja našega algoritma. Na spodnjem primeru si poglejmo ustrezen način delitve tabele.

Primer:

V tabeli imamo števila 12, 5, 6, 17, 8, 24, 11, 25. Števila bomo urejali tako, da si bomo izbrali delilni element, nato pa se bomo postavili z enim indeksom na začetek tabele, z drugim pa na konec tabele. S prvim indeksom se premikamo naprej toliko časa, dokler je vrednost v tabeli manjša od delilnega elementa. Z drugim indeksom pa se pomikamo od zadaj naprej toliko časa, dokler je vrednost v tabeli večja od delilnega elementa. Ko je vrednost v tabeli pri prvem indeksu večja ali enaka delilnemu elementu in vrednost v tabeli pri drugem indeksu manjša ali enaka delilnemu elementu, takrat vrednosti med seboj zamenjamo. To ponavljamo toliko časa, dokler se indeksa ne prekrižata. Nato uporabimo rekurzivni klic podprograma in uredimo na enak način  vsak del posebej. Na koncu le še izpišemo urejene vrednosti.

Primer:

                                                       12 5 6 17 8 24 11 25

                                                       (                       (
Izberemo delilni element, ki je v algoritmu kar prvi element zaporedja. V našem primeru izberemo za delilni element število 12.

                                                             11 5 6 17 8 24 12 25

                                                                    ( (
                                     11 5 6 8                                                   17 24 12 25

                                 del. element 11                                          del. element 17

                             8  5  6                   11                              12               24  17  25

                   delilni element je 8                                                    delilni element je 24

                         6  5             8                                                      17                       24      25

             nov delilni element 6                                                                     nov delilni element je 24                                   
                       5 6                                                                                                      24      25
Na koncu izpišemo 5 6 8 11 12 17 24 25.

program tocke;

type tabela = record

ocena: array[1..100] of integer;

ime : string

end;

var

tab: tabela;

stev,i: integer;

procedure napolni_tabelo(var tab : tabela;stevilo:integer);

{Napolnimo tabelo z doseženimi točkami učencev.}

var

i : integer;

begin

  for i:=1 to stevilo do 

  begin

    write(i:0,' - ti ucenec je ');

    readln(tab.ime[i]);

    write('Njegova ocena je ');

    readln(tab.ocena[i]);

  end;

end;

procedure razvrsti(var pod:tabela;l,d: integer);

{Razvrsti ocene po velikosti.}

var

i,j,delilni,pomozni:  integer;

begin

  i:=l;

  j:=d;

  delilni:=pod.ocena[1];    {delilni element}

  while i<=j do 

  begin

    while ((pod.ocena[i] < delilni) and (i < d)) do i:=i+1;

    while ((pod.ocena[j] > delilni) and (j > l)) do j:=j-1;

    if i<=j then 

    begin

      pomozni:=pod.ocena[i];

      pod.ocena[i]:=pod.ocena[j];

      pod.ocena[j]:=pomozni;

      i:=i + 1;

      j:=j-1;

    end;

  end;

  if (i<d) then razvrsti(pod,i,d)     {rekurzivni klic}

  if (j>l) then razvrsti(pod,l,j);    {rekurzivni klic}

end;

begin {glavni program}

  write('Vnesite stevilo ucencev (maksimalno 100) ');

  readln(stev);

  napolni_tabelo(tab,stev);

  razvrsti(tab,1,stev);

  for i:=1 to stev do

  begin

    writeln(tab.ime[i],'  ',tab.ocena[i]:0);

  end;

end.

Za vajo reši naslednja primera:

1. Reši nalogo 'primer 7' tako, da boš upošteval še imena učencev.

2. Na državnih volitvah za poslance v parlamentu je sodelovalo 1005 kandidatov. Poznano je  število glasov, ki jih je pridobil posamezen kandidat. Napiši program, ki bo razporedil kandidate glede na število dobljenih glasov.

5. primer

Z algoritmom za zlivanje uredi tabelo števil.

Nalogo bomo rešili tako, da bomo najprej tabelo razdelili na dva dela (dve podzaporedji). Nato bomo vsako podzaporedje zopet delili naprej. To bomo počeli tako dolgo, dokler v podzaporedju, ki ga delimo, ne bo samo še en element. To bo naš zaustavitveni pogoj. Nato pa bomo uporabili ustrezen podprogram (imenovali ga bomo zlij), ki bo združeval urejena podzaporedja. Delitev tabele bomo opravili s pomočjo rekurzije, pri čemer bomo za delilni element vzeli kar element, ki je na sredini tabele. 

  Ker bomo zlivali elemente tako, da bomo združevali urejena podzaporedja, bo tudi končno zaporedje urejeno. Podprogram zlij bo namreč pri vsakem zlivanju podzaporedji primerjal elemente enega podzaporedja z elementi drugega podzaporedja in jih zlil v novo urejeno podzaporedje. 
Primer, kako delimo tabelo elementov:

12 5 6 17 8 24 11 25

delilni element je n div 2 je 8 div 2 je 4

12 5 6 17                      8 24 11 25

delilni element je 2                        delilni element je 2

12 5                  6 17                      8 24                11 25

delilni element je 1
12                5                  6                 17                      8            24                11           25
Primer, kako zlivamo tabelo.


program zlivanje;

type podatki = array[1..100] of integer;

var

tab : podatki;

dno, vrh, i  : integer;

procedure napolni_tabelo(var tab : podatki);

{vnos vrednosti v tabelo}

var

i, stevilo : integer;

begin

  write('Vnesite stevilo elementov v tabeli (maksimalno 100) ');

  readln(stevilo);

  for i:=1 to stevilo do begin

    write(i:0,' - to stevilo je ');

    readln(tab[i]);

  end;

end;

procedure zlij(var a: podatki; var dno, s, vrh: integer);

{Podprogram združi urejeni zaporedji v eno zaporedje.}

var

h, j, i, k: integer;

b : podatki;

begin

  h:= dno; j:= s + 1; i := dno; {Indeksi tekočih elementov.}

  while (h <= s) and ( j <= vrh) do

  begin

    if a[h] <= a[j] then {Dokler je v obeh prvotnih zaporedjih

    še kaj elementov, jemljemo v zaporedje manjše elemente}

    begin

      b[i] := a[h]; 

      h := h + 1

      end

    else

    begin

      b[i] := a[j]; j := j + 1

    end; {Novi element je v zaporedju b.}

    i := i + 1

  end;

  {Eno od zaporedij podatkov je izčrpano.}

  if h > s then

  {Ostali so elementi v spodnjem delu a.}

  for k := j to vrh do

  begin

    b[i] := a[k];

    i := i + 1

  end

  else

  {Ostali so elementi v zgornjem delu a.}

  for k := h to s do

  begin

    b[i] := a[k];

    i := i + 1

  end;

  {Preložimo rezultat nazaj v zaporedje a.}

  for k := dno to vrh do a[k] := b[k]

end;

procedure uredi_z_zlivanjem(var a: podatki; var dno, vrh: integer);

{Podprogram uredi z zlivanjem elemente ai, i = dno, dno +1,.., vrh

v nepadajočem vrstnem redu.Veljati mora, da je dno ≤ vrh.}

var

s, s1 : integer;

begin

  if dno < vrh then  {zaustavitveni pogoj}

    begin

    s := (dno + vrh) div 2;

    uredi_z_zlivanjem(a, dno, s);

    s1 := s + 1;

    uredi_z_zlivanjem(a, s1, vrh);

    zlij(a, dno, s, vrh) {{združi rešitev podproblemov}

  end;

end;

begin { glavni program}

  napolni_tabelo(tab); 

  write('Od katerega elementa urejam tabelo ');

  readln(dno);

  write('Do katerega elementa urejam tabelo ');

  readln(vrh);

  uredi_z_zlivanjem(tab, dno, vrh); 

  for i:= dno to vrh do 

  write(tab[i]: 0,',  ');

end.

Za vajo reši naslednji primer:

1. V tabeli imaš 10000 naključnih vrednosti tipa integer. Uredi jih po velikosti.
6. primer

Napiši program z rekurzivno funkcijo A(n,x,y), ki bo izračunal n- ti člen rekurzivnega zaporedja:

a0 = x, a1 = y in an = an-1  + 3 * an-2.

Nalogo bomo rešili tako, da bomo najprej preverili, če je n dovolj majhen(0, 1 ali 2). Če je n dovolj majhen bomo lahko določili n- ti člen zaporedja, sicer pa bo program zapisal podatke na sklad in poklical funkcijo za (n-1). S tem bo nadaljeval tako dolgo, dokler ne bo neki novi n = 2. Ko bo program izračunal vrednost za n = 2 bo vzel iz sklada novi n in izračunal vrednost zaporedja. To bo ponavljal tako dolgo, dokler sklad ne bo prazen.

Primer:

x = 2;

y = 1;

n = 4

Ker ni n =0 ali 1 ali 2 se bodo podatki zapisali na sklad. Izvedel se bo klic funkcije za (n-1), torej n = 3. Ker še vedno niso izpolnjeni pogoji se zopet zapišejo vrednosti na sklad, izvede pa se klic funkcije za (n-1) torej 2. Sedaj  se izračuna vrednost za a2 = a1 + 3*a0 = y + 3*x = 1 + 3*2 = 7. Nato prebere program podatke iz sklada za en nivo višje (n = 3) in izračuna a3 = a2 + 3 * a1  = 7 + 3 * 1 = 10. Ker sklad še ni prazen se izračuna še vrednost za n = 4 in sicer a4 = a3 + 3 * a2 = 10 + 3 * 7 = 31. Sedaj je sklad prazen in funkcija se zaključi.

program zaporedje;

{določi n-ti člen zaporedja}

var

n,x,y: integer;

function del(n,x,y: integer):integer;

begin

   if n = 0 then  begin  del:=x; end              {zaustavitveni pogoj}

   else  if n = 1 then  del:=y                         {zaustavitveni pogoj}

   else  if n=2 then  del:=y + 3 * x

   else  del:=del(n-1,x,y)+3*del(n-2,x,y);          {rekurzivni klic}

end;

begin          {glavni program}

  write('vnesi n');

  readln(n);

  write('vnesi x');

  readln(x);

  write('vnesi y');

  readln(y);

  writeln('N ti clen zaporedja je ',del(n,x,y):3);

end.

Za vajo reši naslednja primera:

1. Napiši program z rekurzivno funkcijo A(n,x), ki bo izračunal n- ti člen rekurzivnega zaporedja:

a0 = x,  in an = 4 * an-1.

2. Napiši program z rekurzivno funkcijo A(n,x,y,z), ki bo izračunal n- ti člen rekurzivnega zaporedja:

a0 = x,  a1 = y, a2 = z in an = an-1 + 2 * an-2 + an-3.
7. primer

Sestavljamo kvadrate. Vsak naslednji kvadrat dobimo tako, da obstoječemu kvadratu priložimo še tri enake kvadrate. Če je stranica prvega kvadrata a izvedemo pa n korakov, ugotovi ploščino nastalega kvadrata.

Nalogo bomo rešili tako, da če je n = 0 bo ploščina novega kvadrata kar a * a. Če pa je n > 0 pa bomo računali ploščino tako, da bomo najprej izračunali ploščino kvadrata za n = 1, 2, ... n. Uporabili bomo torej rekurzivni klic funkcije.

Primer:

n = 2

a = 3

Ker je n > 0 bo program zapisal podatke na sklad in izvedel rekurzivni klic za (n - 1), torej n = 1. Ker je novi n (n = 1) še vedno večji od 0, se bodo podatki zopet zapisali na sklad, n pa se bo zmanjšal za 1. Novi n je sedaj enak 0 zato se izračuna ploščina kvadrata p0 = a * a = 3 * 3 = 9. Vzamejo se podatki iz sklada za n = 1 in izračuna p1 = 4 * 9 = 36. Ker sklad še ni prazen se prejšnji korak ponovi za n = 2 in izračuna p2 = 4 * 36 = 144. 
program kvadrat;

{Določi ploščino kvadrata po n korakih, ki ga dobimo tako,

da se vsak korak poveča osnovna ploščina za 4- krat}

var

n: integer;

a: real;

function ploscina(n: integer; a: real):real;

var

p: real;

begin

  if n= 0 then              {zaustavitveni pogoj}

  p:=a*a

  else

  p:=4*ploscina(n-1,a);      {rekurzivni klic}

  ploscina:=p;

end;

begin {glavni program}

  write('Vnesite velikost stranice a ');

  readln(a);

  write('Stevilo korakov n je ');

  readln(n);

  writeln;

  writeln('Ploscina kvadrata je ',ploscina(n,a):3:0);

end.

Za vajo reši naslednja primera:

1. Imamo kocko z osnovnim robom a. Na vsakemu koraku dodamo kocki še 7 enakih kock in tako dobimo novo kocko. Določi prostornino tako nastale kocke po n korakih.

2. Imamo krog s polmerom r. Iz tega kroga izrežemo nov krog, ki ima polmer r/2. Določi ploščino tako nastalega kroga po n korakih!

